

NAG Library Routine Document

F08HPF (ZHBEVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08HPF (ZHBEVX) computes selected eigenvalues and, optionally, eigenvectors of a complex n by n Hermitian band matrix A of bandwidth $(2k_d + 1)$. Eigenvalues and eigenvectors can be selected by specifying either a range of values or a range of indices for the desired eigenvalues.

2 Specification

```
SUBROUTINE F08HPF (JOBZ, RANGE, UPLO, N, KD, AB, LDAB, Q, LDQ, VL, VU,      &
                  IL, IU, ABSTOL, M, W, Z, LDZ, WORK, RWORK, IWORK,      &
                  JFAIL, INFO)

INTEGER             N, KD, LDAB, LDQ, IL, IU, M, LDZ, IWORK(5*N),      &
                   JFAIL(*), INFO
REAL (KIND=nag_wp) VL, VU, ABSTOL, W(N), RWORK(7*N)
COMPLEX (KIND=nag_wp) AB(LDAB,*), Q(LDQ,*), Z(LDZ,*), WORK(N)
CHARACTER(1)        JOBZ, RANGE, UPLO
```

The routine may be called by its LAPACK name *zhbevx*.

3 Description

The Hermitian band matrix A is first reduced to real tridiagonal form, using unitary similarity transformations. The required eigenvalues and eigenvectors are then computed from the tridiagonal matrix; the method used depends upon whether all, or selected, eigenvalues and eigenvectors are required.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- | | |
|--|--|
| 1: JOBZ – CHARACTER(1) | <i>Input</i> |
| On entry: indicates whether eigenvectors are computed. | |
| JOBZ = 'N' | Only eigenvalues are computed. |
| JOBZ = 'V' | Eigenvalues and eigenvectors are computed. |
| Constraint: JOBZ = 'N' or 'V'. | |

2:	RANGE – CHARACTER(1)	<i>Input</i>
<i>On entry:</i> if RANGE = 'A', all eigenvalues will be found.		
If RANGE = 'V', all eigenvalues in the half-open interval (VL, VU] will be found.		
If RANGE = 'I', the ILth to IUth eigenvalues will be found.		
<i>Constraint:</i> RANGE = 'A', 'V' or 'I'.		
3:	UPLO – CHARACTER(1)	<i>Input</i>
<i>On entry:</i> if UPLO = 'U', the upper triangular part of A is stored.		
If UPLO = 'L', the lower triangular part of A is stored.		
<i>Constraint:</i> UPLO = 'U' or 'L'.		
4:	N – INTEGER	<i>Input</i>
<i>On entry:</i> n, the order of the matrix A.		
<i>Constraint:</i> N ≥ 0.		
5:	KD – INTEGER	<i>Input</i>
<i>On entry:</i> if UPLO = 'U', the number of superdiagonals, k_d , of the matrix A.		
If UPLO = 'L', the number of subdiagonals, k_d , of the matrix A.		
<i>Constraint:</i> KD ≥ 0.		
6:	AB(LDAB,*) – COMPLEX (KIND=nag_wp) array	<i>Input/Output</i>
Note: the second dimension of the array AB must be at least max(1,N).		
<i>On entry:</i> the upper or lower triangle of the n by n Hermitian band matrix A.		
The matrix is stored in rows 1 to $k_d + 1$, more precisely,		
if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in AB($k_d + 1 + i - j, j$) for $\max(1, j - k_d) \leq i \leq j$;		
if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in AB($1 + i - j, j$) for $j \leq i \leq \min(n, j + k_d)$.		
<i>On exit:</i> AB is overwritten by values generated during the reduction to tridiagonal form.		
The first superdiagonal or subdiagonal and the diagonal of the tridiagonal matrix T are returned in AB using the same storage format as described above.		
7:	LDAB – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array AB as declared in the (sub)program from which F08HPF (ZHBEVX) is called.		
<i>Constraint:</i> LDAB ≥ KD + 1.		
8:	Q(LDQ,*) – COMPLEX (KIND=nag_wp) array	<i>Output</i>
Note: the second dimension of the array Q must be at least max(1,N) if JOBZ = 'V', and at least 1 otherwise.		
<i>On exit:</i> if JOBZ = 'V', the n by n unitary matrix used in the reduction to tridiagonal form.		
If JOBZ = 'N', Q is not referenced.		

9: LDQ – INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F08HPF (ZHBEVX) is called.

Constraints:

if $\text{JOBZ} = \text{'V'}$, $\text{LDQ} \geq \max(1, N)$;
 otherwise $\text{LDQ} \geq 1$.

10: VL – REAL (KIND=nag_wp) *Input*

11: VU – REAL (KIND=nag_wp) *Input*

On entry: if $\text{RANGE} = \text{'V'}$, the lower and upper bounds of the interval to be searched for eigenvalues.

If $\text{RANGE} = \text{'A'}$ or 'T' , VL and VU are not referenced.

Constraint: if $\text{RANGE} = \text{'V'}$, $\text{VL} < \text{VU}$.

12: IL – INTEGER *Input*

13: IU – INTEGER *Input*

On entry: if $\text{RANGE} = \text{'I'}$, the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If $\text{RANGE} = \text{'A'}$ or 'V' , IL and IU are not referenced.

Constraints:

if $\text{RANGE} = \text{'I'}$ and $N = 0$, $\text{IL} = 1$ and $\text{IU} = 0$;
 if $\text{RANGE} = \text{'I'}$ and $N > 0$, $1 \leq \text{IL} \leq \text{IU} \leq N$.

14: ABSTOL – REAL (KIND=nag_wp) *Input*

On entry: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval $[a, b]$ of width less than or equal to

$$\text{ABSTOL} + \epsilon \max(|a|, |b|),$$

where ϵ is the **machine precision**. If ABSTOL is less than or equal to zero, then $\epsilon \|T\|_1$ will be used in its place, where T is the tridiagonal matrix obtained by reducing A to tridiagonal form. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold $2 \times \text{X02AMF}()$, not zero. If this routine returns with $\text{INFO} > 0$, indicating that some eigenvectors did not converge, try setting ABSTOL to $2 \times \text{X02AMF}()$. See Demmel and Kahan (1990).

15: M – INTEGER *Output*

On exit: the total number of eigenvalues found. $0 \leq M \leq N$.

If $\text{RANGE} = \text{'A'}$, $M = N$.

If $\text{RANGE} = \text{'I'}$, $M = \text{IU} - \text{IL} + 1$.

16: W(N) – REAL (KIND=nag_wp) array *Output*

On exit: the first M elements contain the selected eigenvalues in ascending order.

17: Z(LDZ, *) – COMPLEX (KIND=nag_wp) array *Output*

Note: the second dimension of the array Z must be at least $\max(1, M)$ if $\text{JOBZ} = \text{'V'}$, and at least 1 otherwise.

On exit: if $\text{JOBZ} = \text{'V'}$, then

if $\text{INFO} = 0$, the first M columns of Z contain the orthonormal eigenvectors of the matrix A corresponding to the selected eigenvalues, with the i th column of Z holding the eigenvector associated with $\mathbf{W}(i)$;

if an eigenvector fails to converge ($\text{INFO} > 0$), then that column of Z contains the latest approximation to the eigenvector, and the index of the eigenvector is returned in JFAIL .

If $\text{JOBZ} = \text{'N'}$, Z is not referenced.

Note: you must ensure that at least $\max(1, M)$ columns are supplied in the array Z ; if $\text{RANGE} = \text{'V'}$, the exact value of M is not known in advance and an upper bound of at least N must be used.

18: $\text{LDZ} - \text{INTEGER}$

Input

On entry: the first dimension of the array Z as declared in the (sub)program from which F08HPF (ZHBEVX) is called.

Constraints:

if $\text{JOBZ} = \text{'V'}$, $\text{LDZ} \geq \max(1, N)$;
 otherwise $\text{LDZ} \geq 1$.

19: $\text{WORK}(N) - \text{COMPLEX} (\text{KIND}=\text{nag_wp})$ array

Workspace

20: $\text{RWORK}(7 \times N) - \text{REAL} (\text{KIND}=\text{nag_wp})$ array

Workspace

21: $\text{IWORK}(5 \times N) - \text{INTEGER}$ array

Workspace

22: $\text{JFAIL}(\text{*}) - \text{INTEGER}$ array

Output

Note: the dimension of the array JFAIL must be at least $\max(1, N)$.

On exit: if $\text{JOBZ} = \text{'V'}$, then

if $\text{INFO} = 0$, the first M elements of JFAIL are zero;
 if $\text{INFO} > 0$, JFAIL contains the indices of the eigenvectors that failed to converge.

If $\text{JOBZ} = \text{'N'}$, JFAIL is not referenced.

23: $\text{INFO} - \text{INTEGER}$

Output

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

If $\text{INFO} = i$, then i eigenvectors failed to converge. Their indices are stored in array JFAIL . Please see ABSTOL.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F08HPF (ZHBEVX) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08HPF (ZHBEVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to k_dn^2 if $\text{JOBZ} = \text{'N'}$, and is proportional to n^3 if $\text{JOBZ} = \text{'V'}$ and $\text{RANGE} = \text{'A'}$, otherwise the number of floating-point operations will depend upon the number of computed eigenvectors.

The real analogue of this routine is F08HBF (DSBEVX).

10 Example

This example finds the eigenvalues in the half-open interval $(-2, 2]$, and the corresponding eigenvectors, of the Hermitian band matrix

$$A = \begin{pmatrix} 1 & 2-i & 3-i & 0 & 0 \\ 2+i & 2 & 3-2i & 4-2i & 0 \\ 3+i & 3+2i & 3 & 4-3i & 5-3i \\ 0 & 4+2i & 4+3i & 4 & 5-4i \\ 0 & 0 & 5+3i & 5+4i & 5 \end{pmatrix}.$$

10.1 Program Text

```
Program f08hpfe
!
!     F08HPF Example Program Text
!
!     Mark 26 Release. NAG Copyright 2016.
!
!     .. Use Statements ..
Use nag_library, Only: nag_wp, x04daf, zhbevx
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Real (Kind=nag_wp), Parameter      :: zero = 0.0E+0_nag_wp
Integer, Parameter                 :: nin = 5, nout = 6
Character (1), Parameter          :: uplo = 'U'
!     .. Local Scalars ..
Real (Kind=nag_wp)                  :: abstol, vl, vu
Integer                           :: i, ifail, il, info, iu, j, kd, ldab, &
                                    ldq, ldz, m, n
!     .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: ab(:,:), q(:,:), work(:, ), z(:,:)
Real (Kind=nag_wp), Allocatable   :: rwork(:, ), w(:, )
Integer, Allocatable             :: iwork(:, ), jfail(:, )
!     .. Intrinsic Procedures ..
Intrinsic                          :: max, min
!     .. Executable Statements ..
Write (nout,*) 'F08HPF Example Program Results'
Write (nout,*) 
!     Skip heading in data file
Read (nin,*) 
Read (nin,*) n, kd
ldab = kd + 1
ldq = n
```

```

ldz = n
m = n
Allocate (ab(ldab,n),q(ldq,n),work(n),z(ldz,m),rwork(7*n),w(n),
          iwork(5*n),jfail(n)) &

!      Read the lower and upper bounds of the interval to be searched,
!      and read the upper or lower triangular part of the matrix A
!      from data file

Read (nin,*) vl, vu
If (uplo=='U') Then
    Read (nin,*)((ab(kd+1+i-j,j),j=i,min(n,i+kd)),i=1,n)
Else If (uplo=='L') Then
    Read (nin,*)((ab(1+i-j,j),j=max(1,i-kd),i),i=1,n)
End If

!      Set the absolute error tolerance for eigenvalues. With ABSTOL
!      set to zero, the default value is used instead

abstol = zero

!      Solve the band symmetric eigenvalue problem
!      The NAG name equivalent of zhbevx is f08hpfe
Call zhbevx('Vectors','Values in range',uplo,n,kd,ab,ldab,q,ldq,vl,vu,   &
            il,iu,abstol,m,w,z,ldz,work,rwork,iwork,jfail,info)

If (info>=0) Then

!      Print solution

Write (nout,99999) 'Number of eigenvalues found =', m
Write (nout,*)
Write (nout,*) 'Eigenvalues'
Write (nout,99998) w(1:m)
Flush (nout)

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04daf('General',' ',n,m,z,ldz,'Selected eigenvectors',ifail)

If (info>0) Then
    Write (nout,99999) 'INFO eigenvectors failed to converge, INFO =',   &
                      info
    Write (nout,*) 'Indices of eigenvectors that did not converge'
    Write (nout,99997) jfail(1:m)
End If
Else
    Write (nout,99999) 'Failure in ZHBEVX. INFO =', info
End If

99999 Format (1X,A,I5)
99998 Format (3X,(8F8.4))
99997 Format (3X,(8I8))
End Program f08hpfe

```

10.2 Program Data

F08HPF Example Program Data

```

5           2                               :Values of N and KD
-2.0        2.0                            :Values of VL and VU
(1.0, 0.0) (2.0,-1.0) (3.0,-1.0)
(2.0, 0.0) (3.0,-2.0) (4.0,-2.0)
(3.0, 0.0) (4.0,-3.0) (5.0,-3.0)
(4.0, 0.0) (5.0,-4.0)
(5.0, 0.0) :End of matrix A

```

10.3 Program Results

F08HPF Example Program Results

Number of eigenvalues found = 2

Eigenvalues

-1.4094 1.4421

Selected eigenvectors

1 2

1 0.6367 0.4516
0.0000 0.0000

2 -0.2578 -0.3029
0.2413 -0.4402

3 -0.3039 0.3160
-0.3481 0.2978

4 0.3450 -0.4088
-0.0832 -0.3213

5 -0.2469 0.0204
0.2634 0.2262
